NAG Library Routine Document

G13DMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13DMF calculates the sample cross-correlation (or cross-covariance) matrices of a multivariate time series.

2 Specification

```
SUBROUTINE G13DMF (MATRIX, K, N, M, W, KMAX, WMEAN, R0, R, IFAIL)

INTEGER            K, N, M, KMAX, IFAIL
REAL (KIND=nag_wp) W(KMAX,N), WMEAN(K), R0(KMAX,K), R(KMAX,KMAX,M)
CHARACTER(1)       MATRIX
```

3 Description

Let $W_t = (w_{1t}, w_{2t}, \ldots, w_{kt})^{\mathrm{T}}$, for $t = 1, 2, \ldots, n$, denote n observations of a vector of k time series. The sample cross-covariance matrix at lag l is defined to be the k by k matrix $\hat{C}(l)$, whose (i, j)th element is given by

$$\hat{C}_{ij}(l) = \frac{1}{n} \sum_{t=l+1}^{n} \left(w_{i(t-l)} - \bar{w}_i\right)\left(w_{jt} - \bar{w}_j\right), \qquad l = 0, 1, 2, \ldots, m,\ i = 1, 2, \ldots, k \text{ and } j = 1, 2, \ldots, k,$$

where $\bar{w}_i$ and $\bar{w}_j$ denote the sample means for the ith and jth series respectively. The sample cross-correlation matrix at lag l is defined to be the k by k matrix $\hat{R}(l)$, whose (i, j)th element is given by

$$\hat{R}_{ij}(l) = \frac{\hat{C}_{ij}(l)}{\sqrt{\hat{C}_{ii}(0)\hat{C}_{jj}(0)}}, \qquad l = 0, 1, 2, \ldots, m,\ i = 1, 2, \ldots, k \text{ and } j = 1, 2, \ldots, k.$$

The number of lags, m, is usually taken to be at most $n/4$.

If W_t follows a vector moving average model of order q, then it can be shown that the theoretical cross-correlation matrices $(R(l))$ are zero beyond lag q. In order to help spot a possible cut-off point, the elements of $\hat{R}(l)$ are usually compared to their approximate standard error of $1/\sqrt{n}$. For further details see, for example, Wei (1990).

The routine uses a single pass through the data to compute the means and the cross-covariance matrix at lag zero. The cross-covariance matrices at further lags are then computed on a second pass through the data.

4 References

Wei W W S (1990) *Time Series Analysis: Univariate and Multivariate Methods* Addison–Wesley

West D H D (1979) Updating mean and variance estimates: An improved method *Comm. ACM* **22** 532–555

5 Parameters

1: MATRIX – CHARACTER(1) *Input*

On entry: indicates whether the cross-covariance or cross-correlation matrices are to be computed.

MATRIX = 'V'
: The cross-covariance matrices are computed.

MATRIX = 'R'
: The cross-correlation matrices are computed.

Constraint: MATRIX = 'V' or 'R'.

2: K – INTEGER *Input*

On entry: k, the dimension of the multivariate time series.

Constraint: $\mathrm{K} \geq 1$.

3: N – INTEGER *Input*

On entry: n, the number of observations in the series.

Constraint: $\mathrm{N} \geq 2$.

4: M – INTEGER *Input*

On entry: m, the number of cross-correlation (or cross-covariance) matrices to be computed. If in doubt set $\mathrm{M} = 10$. However it should be noted that M is usually taken to be at most $\mathrm{N}/4$.

Constraint: $1 \leq \mathrm{M} < \mathrm{N}$.

5: W(KMAX,N) – REAL (KIND=nag_wp) array *Input*

On entry: $\mathrm{W}(i,t)$ must contain the observation w_{it}, for $i = 1, 2, \ldots, k$ and $t = 1, 2, \ldots, n$.

6: KMAX – INTEGER *Input*

On entry: the first dimension of the arrays W, R0 and R and the second dimension of the array R as declared in the (sub)program from which G13DMF is called.

Constraint: $\mathrm{KMAX} \geq \mathrm{K}$.

7: WMEAN(K) – REAL (KIND=nag_wp) array *Output*

On exit: the means, $\bar{w}_i$, for $i = 1, 2, \ldots, k$.

8: R0(KMAX,K) – REAL (KIND=nag_wp) array *Output*

On exit: if $i \neq j$, then $\mathrm{R0}(i,j)$ contains an estimate of the (i,j)th element of the cross-correlation (or cross-covariance) matrix at lag zero, $\hat{R}_{ij}(0)$; if $i = j$, then if MATRIX = 'V', $\mathrm{R0}(i,i)$ contains the variance of the ith series, $\hat{C}_{ii}(0)$, and if MATRIX = 'R', $\mathrm{R0}(i,i)$ contains the standard deviation of the ith series, $\sqrt{\hat{C}_{ii}(0)}$.

If $\mathrm{IFAIL} = 2$ and MATRIX = 'R', then on exit all the elements in R0 whose computation involves the zero variance are set to zero.

9: R(KMAX,KMAX,M) – REAL (KIND=nag_wp) array *Output*

On exit: $\mathrm{R}(i,j,l)$ contains an estimate of the (i,j)th element of the cross-correlation (or cross-covariance) at lag l, $\hat{R}_{ij}(l)$, for $l = 1, 2, \ldots, m$, $i = 1, 2, \ldots, k$ and $j = 1, 2, \ldots, k$.

If $\mathrm{IFAIL} = 2$ and MATRIX = 'R', then on exit all the elements in R whose computation involves the zero variance are set to zero.

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, MATRIX $\neq$ 'V' or 'R',
or K < 1,
or N < 2,
or M < 1,
or M $\geq$ N,
or KMAX $<$ K.

IFAIL $= 2$

On entry, at least one of the k series is such that all its elements are practically equal giving zero (or near zero) variance. In this case if MATRIX = 'R' all the correlations in R0 and R involving this variance are set to zero.

7 Accuracy

For a discussion of the accuracy of the one-pass algorithm used to compute the sample cross-covariances at lag zero see West (1979). For the other lags a two-pass algorithm is used to compute the cross-covariances; the accuracy of this algorithm is also discussed in West (1979). The accuracy of the cross-correlations will depend on the accuracy of the computed cross-covariances.

8 Further Comments

The time taken is roughly proportional to mnk^2.

9 Example

This program computes the sample cross-correlation matrices of two time series of length 48, up to lag 10. It also prints the cross-correlation matrices together with plots of symbols indicating which elements of the correlation matrices are significant. Three * represent significance at the 0.5% level, two * represent significance at the 1% level and a single * represents significance at the 5% level. The * are plotted above or below the line depending on whether the elements are significant in the positive or negative direction.

9.1 Program Text

```
!   G13DMF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.

    Module g13dmfe_mod

!     G13DMF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                   :: nin = 5, nout = 6
    Contains
      Subroutine cprint(k,n,ldr,m,wmean,r0,r,nout)

!       .. Use Statements ..
        Use nag_library, Only: x04cbf
!       .. Scalar Arguments ..
        Integer, Intent (In)                 :: k, ldr, m, n, nout
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (In)      :: r(ldr,ldr,m), r0(ldr,k),       &
                                                wmean(k)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                   :: c1, c2, c3, c5, c6, c7,        &
                                                inv_sqrt_n, sum
        Integer                              :: i, i2, ifail, j, l, ll
!       .. Local Arrays ..
        Character (1)                        :: clabs(1), rlabs(1)
        Character (80)                       :: rec(7)
!       .. Intrinsic Procedures ..
        Intrinsic                            :: real, sqrt
!       .. Executable Statements ..
!       Print the correlation matrices and indicator symbols.

        inv_sqrt_n = 1.0E0_nag_wp/sqrt(real(n,kind=nag_wp))
        Write (nout,*)
        Write (nout,*) ' THE MEANS'
        Write (nout,*) ' ---------'
        Write (nout,99999) wmean(1:k)
        Write (nout,*)
        Write (nout,*) ' CROSS-CORRELATION MATRICES'
        Write (nout,*) ' --------------------------'
        Write (nout,99998) ' Lag = ', 0
        Flush (nout)
        ifail = 0
        Call x04cbf('G','N',k,k,r0,ldr,'F9.3',' ','N',rlabs,'N',clabs,80,5, &
          ifail)
        Do l = 1, m
          Write (nout,99998) ' Lag = ', l
          Flush (nout)
          ifail = 0
          Call x04cbf('G','N',k,k,r(1,1,l),ldr,'F9.3',' ','N',rlabs,'N',clabs, &
            80,5,ifail)
        End Do

!       Print indicator symbols to indicate significant elements.
        Write (nout,99997) ' Standard error = 1 / SQRT(N) =', inv_sqrt_n
        Write (nout,*)
        Write (nout,*) ' TABLES OF INDICATOR SYMBOLS'
        Write (nout,*) ' ---------------------------'
        Write (nout,99998) ' For Lags 1 to ', m

!       Set up annotation for the plots.
        Write (rec(1),99996) '                0.005  :'
        Write (rec(2),99996) '        +       0.01   :'
        Write (rec(3),99996) '                0.05   :'
        Write (rec(4)(1:23),99996) '   Sig. Level        :'
```

```
        Write (rec(4)(24:),99996) '- - - - - - - - - -  Lags'
        Write (rec(5),99996) '              0.05   :'
        Write (rec(6),99996) '        -     0.01   :'
        Write (rec(7),99996) '              0.005  :'

!       Set up the critical values
        c1 = 3.29E0_nag_wp*inv_sqrt_n
        c2 = 2.58E0_nag_wp*inv_sqrt_n
        c3 = 1.96E0_nag_wp*inv_sqrt_n
        c5 = -c3
        c6 = -c2
        c7 = -c1

        Do i = 1, k
          Do j = 1, k
            Write (nout,*)
            If (i==j) Then
              Write (nout,99995) ' Auto-correlation function for', ' series ', &
                i
            Else
              Write (nout,99994) ' Cross-correlation function for', &
                ' series ', i, ' and series', j
            End If
            Do l = 1, m
              ll = 23 + 2*l
              sum = r(i,j,l)

!             Clear the last plot with blanks
              Do i2 = 1, 7
                If (i2/=4) Then
                  rec(i2)(ll:ll) = ' '
                End If
              End Do

!             Check for significance
              If (sum>c1) Then
                rec(1)(ll:ll) = '*'
              End If
              If (sum>c2) Then
                rec(2)(ll:ll) = '*'
              End If
              If (sum>c3) Then
                rec(3)(ll:ll) = '*'
              End If
              If (sum<c5) Then
                rec(5)(ll:ll) = '*'
              End If
              If (sum<c6) Then
                rec(6)(ll:ll) = '*'
              End If
              If (sum<c7) Then
                rec(7)(ll:ll) = '*'
              End If
            End Do

!           Print
            Write (nout,99996)(rec(i2),i2=1,7)
          End Do
        End Do
        Return

99999   Format (/1X,2(2X,F9.3))
99998   Format (/1X,A,I2)
99997   Format (/1X,A,F6.3,A)
99996   Format (1X,A)
99995   Format (//1X,A,A,I2/)
99994   Format (//1X,A,A,I2,A,I2/)
      End Subroutine cprint
    End Module g13dmfe_mod
    Program g13dmfe
```

```
!     G13DMF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: g13dmf, nag_wp
      Use g13dmfe_mod, Only: cprint, nin, nout
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Integer                              :: i, ifail, k, kmax, m, n
      Character (1)                        :: matrix
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable      :: r(:,:,:), r0(:,:), w(:,:),       &
                                              wmean(:)
!     .. Executable Statements ..
      Write (nout,*) 'G13DMF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) k, n, m, matrix

      kmax = k
      Allocate (w(kmax,n),r0(kmax,k),wmean(k),r(kmax,kmax,m))

!     Read in series
      Do i = 1, k
        Read (nin,*) w(i,1:n)
      End Do

!     Calculate sample cross-correlation matrices
      ifail = 0
      Call g13dmf(matrix,k,n,m,w,kmax,wmean,r0,r,ifail)

!     Display results
      Call cprint(k,n,kmax,m,wmean,r0,r,nout)

    End Program g13dmfe
```

9.2 Program Data

```
G13DMF Example Program Data
2 48 10 'R'                                                 ::  K,N,M,MATRIX
-1.490 -1.620  5.200  6.230  6.210  5.860  4.090  3.180
 2.620  1.490  1.170  0.850 -0.350  0.240  2.440  2.580
 2.040  0.400  2.260  3.340  5.090  5.000  4.780  4.110
 3.450  1.650  1.290  4.090  6.320  7.500  3.890  1.580
 5.210  5.250  4.930  7.380  5.870  5.810  9.680  9.070
 7.290  7.840  7.550  7.320  7.970  7.760  7.000  8.350
 7.340  6.350  6.960  8.540  6.620  4.970  4.550  4.810
 4.750  4.760 10.880 10.010 11.620 10.360  6.400  6.240
 7.930  4.040  3.730  5.600  5.350  6.810  8.270  7.680
 6.650  6.080 10.250  9.140 17.750 13.300  9.630  6.800
 4.080  5.060  4.940  6.650  7.940 10.760 11.890  5.850
 9.010  7.500 10.020 10.380  8.150  8.370 10.730 12.140 :: End of W
```

9.3 Program Results

```
 G13DMF Example Program Results


  THE MEANS
  ---------

       4.370      7.868

  CROSS-CORRELATION MATRICES
  --------------------------
```

```
Lag =  0
         2.818    0.249
         0.249    2.815

Lag =  1
         0.736    0.174
         0.211    0.555

Lag =  2
         0.456    0.076
         0.069    0.260

Lag =  3
         0.379    0.014
         0.026   -0.038

Lag =  4
         0.322    0.110
         0.093   -0.236

Lag =  5
         0.341    0.269
         0.087   -0.250

Lag =  6
         0.363    0.344
         0.132   -0.227

Lag =  7
         0.280    0.425
         0.207   -0.128

Lag =  8
         0.248    0.522
         0.197   -0.085

Lag =  9
         0.240    0.266
         0.254    0.075

Lag = 10
         0.162   -0.020
         0.267    0.005

Standard error = 1 / SQRT(N) = 0.144

TABLES OF INDICATOR SYMBOLS
---------------------------

For Lags 1 to 10


Auto-correlation function for series  1

              0.005  : *
        +     0.01   : * * *
              0.05   : * * * * * *
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :
        -     0.01   :
              0.005  :


Cross-correlation function for series  1 and series 2

              0.005  :              *
        +     0.01   :            * *
              0.05   :          * * *
   Sig. Level        : - - - - - - - - - -  Lags
```

```
             0.05   :
        -    0.01   :
             0.005  :


 Cross-correlation function for series  2 and series 1

             0.005  :
        +    0.01   :
             0.05   :
    Sig. Level      : - - - - - - - - - - -  Lags
             0.05   :
        -    0.01   :
             0.005  :


 Auto-correlation function for series  2

             0.005  : *
        +    0.01   : *
             0.05   : *
    Sig. Level      : - - - - - - - - - - -  Lags
             0.05   :
        -    0.01   :
             0.005  :
```